

# NAG Fortran Library Routine Document

## G13BEF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G13BEF fits a multi-input model relating one output series to the input series with a choice of three different estimation criteria: nonlinear least-squares, exact likelihood and marginal likelihood. When no input series are present, G13BEF fits a univariate ARIMA model.

### 2 Specification

```

SUBROUTINE G13BEF(MR, NSER, MT, PARA, NPARA, KFC, NXXY, XXY, IXXY, KEF,
1          NIT, KZSP, ZSP, ITC, SD, CM, ICM, S, D, NDF, KZEF,
2          RES, STTF, ISTTF, NSTTF, WA, IWA, MWA, IMWA, KPRIV,
3          IFAIL)
  INTEGER      MR(7), NSER, MT(4,NSER), NPARA, KFC, NXXY, IXXY, KEF,
1          NIT, KZSP, ITC, ICM, NDF, KZEF, ISTTF, NSTTF, IWA,
2          MWA(IMWA), IMWA, KPRIV, IFAIL
  real      PARA(NPARA), XXY(IXXY,NSER), ZSP(4), SD(NPARA),
1          CM(ICM,NPARA), S, D, RES(NXXY), STTF(ISTTF), WA(IWA)

```

### 3 Description

#### 3.1 The Multi-input Model

The output series  $y_t$ , for  $t = 1, 2, \dots, n$ , is assumed to be the sum of (unobserved) components  $z_{i,t}$  which are due respectively to the inputs  $x_{i,t}$ , for  $i = 1, 2, \dots, m$ .

Thus  $y_t = z_{1,t} + \dots + z_{m,t} + n_t$  where  $n_t$  is the error, or output noise component.

A typical component  $z_t$  may be either

- a simple regression component,  $z_t = \omega x_t$  (here  $x_t$  is called a simple input), or
- a transfer function model component which allows for the effect of lagged values of the variable, related to  $x_t$  by

$$z_t = \delta_1 z_{t-1} + \delta_2 z_{t-2} + \dots + \delta_p z_{t-p} + \omega_0 x_{t-b} - \omega_1 x_{t-b-1} - \dots - \omega_q x_{t-b-q}.$$

The noise  $n_t$  is assumed to follow a (possibly seasonal) ARIMA model, i.e., may be represented in terms of an uncorrelated series,  $a_t$ , by the hierarchy of equations

- $\nabla^d \nabla_s^D n_t = c + w_t$
- $w_t = \Phi_1 w_{t-s} + \Phi_2 w_{t-2s} + \dots + \Phi_P w_{t-Ps} + e_t - \Theta_1 e_{t-s} - \Theta_2 e_{t-2s} - \dots - \Theta_Q e_{t-Qs}$
- $e_t = \phi_1 e_{t-1} + \phi_2 e_{t-2} + \dots + \phi_p e_{t-p} + a_t - \theta_1 a_{t-1} - \theta_2 a_{t-2} - \dots - \theta_q a_{t-q}$

as outlined in Section 3 of the document for G13AEF.

**Note:** the orders  $p, q$  appearing in each of the transfer function models and the ARIMA model are not necessarily the same;  $\nabla^d \nabla_s^D n_t$  is the result of applying non-seasonal differencing of order  $d$  and seasonal differencing of seasonality  $s$  and order  $D$  to the series  $n_t$ : the differenced series is then of length  $N = n - d - s \times D$ ; the constant term parameter  $c$  may optionally be held fixed at its initial value (usually, but not necessarily zero) rather than being estimated.

For the purpose of defining an estimation criterion it is assumed that the series  $a_t$  is a sequence of independent Normal variates having mean 0 and variance  $\sigma_a^2$ . An allowance has to be made for the effects

of unobserved data prior to the observation period. For the noise component an allowance is always made using a form of backforecasting.

For each transfer function input, the user has to decide what values are to be assumed for the pre-period terms  $z_0, z_{-1}, \dots, z_{1-p}$  and  $x_0, x_{-1}, \dots, x_{1-b-q}$  which are in theory necessary to re-create the component series  $z_1, z_2, \dots, z_n$ , during the estimation procedure.

The first choice is to assume that all these values are zero. In this case, in order to avoid undesirable transient distortion of the early values  $z_1, z_2, \dots$ , the user is advised first to correct the input series  $x_t$  by subtracting from all the terms a suitable constant to make the early values  $x_1, x_2, \dots$ , close to zero. The series mean  $\bar{x}$  is one possibility, but for a series with strong trend the constant might be simply  $x_1$ .

The second choice is to treat the unknown pre-period terms as nuisance parameters and estimate them along with the other parameters. This choice should be used with caution. For example, if  $p = 1$  and  $b = q = 0$ , it is equivalent to fitting to the data a decaying geometric curve of the form  $A\delta^t$ , for  $t = 1, 2, 3, \dots$ , along with the other inputs, this being the form of the transient. If the output  $y_t$  contains a strong trend of this form, which is not otherwise represented in the model, it will have a tendency to influence the estimate of  $\delta$  away from the value appropriate to the transfer function model.

In most applications the first choice should be adequate, with the option possibly being used as a refinement at the end of the modelling process. The number of nuisance parameters is then  $\max(p, b + q)$ , with a corresponding loss of degrees of freedom in the residuals. If the user aligns the input  $x_t$  with the output by using in its place the shifted series  $x_{t-b}$ , then setting  $b = 0$  in the transfer function model, there is some improvement in efficiency. On some occasions when the model contains two or more inputs, each with estimation of pre-period nuisance parameters, these parameters may be co-linear and lead to failure of the routine. The option must then be 'switched off' for one or more inputs.

### 3.2 The Estimation Criterion

This is a measure of how well a proposed set of parameters in the transfer function and noise ARIMA models matches the data. The estimation routine searches for parameter values which minimize this criterion. For a proposed set of parameter values it is derived by calculating

- (i) the components  $z_{1,t}, z_{2,t}, \dots, z_{m,t}$  as the responses to the input series  $x_{1,t}, x_{2,t}, \dots, x_{m,t}$  using the equations (a) or (b) above,
- (ii) the discrepancy between the output and the sum of these components, as the noise

$$n_t = y_t - (z_{1,t} + z_{2,t} + \dots + z_{m,t}),$$

- (iii) the residual series  $a_t$  from  $n_t$  by reversing the recursive equations (c), (d) and (e) above.

This last step again requires treatment of the effect of unknown pre-period values of  $n_t$  and other terms in the equations regenerating  $a_t$ . This is identical to the treatment given in Section 3 of the document for G13AEF, and leads to a criterion which is a sum of squares function  $S$ , of the residuals  $a_t$ . It may be shown that the finite algorithm presented there is equivalent to taking the infinite set of past values  $n_0, n_{-1}, n_{-2}, \dots$ , as (linear) nuisance parameters. There is no loss of degrees of freedom however, because the sum of squares function  $S$  may be expressed as including the corresponding set of past residuals; see page 273 of Box and Jenkins (1976), who prove that

$$S = \sum_{-\infty}^n a_t^2.$$

The function  $D = S$  is the first of the three possible criteria, and is quite adequate for moderate to long series with no seasonal parameters. The second is the exact likelihood criterion which considers the past set  $n_0, n_{-1}, n_{-2}$  not as simple nuisance parameters, but as unobserved random variables with known distribution. Calculation of the likelihood of the observed set  $n_1, n_2, \dots, n_n$  requires theoretical integration over the range of the past set. Fortunately this yields a criterion of the form  $D = M \times S$  (whose minimization is equivalent to maximizing the exact likelihood of the data), where  $S$  is exactly as before, and the multiplier  $M$  is a function calculated from the ARIMA model parameters. The value of  $M$  is always  $\geq 1$ , and  $M$  tends to 1 for any fixed parameter set as the sample size  $n$  tends to  $\infty$ . There is a moderate computational overhead in using this option, but its use avoids appreciable bias in the ARIMA model parameters and yields a better conditioned estimation problem.

The third criterion of marginal likelihood treats the coefficients of the simple inputs in a manner analogous to that given to the past set  $n_0, n_{-1}, n_{-2}, \dots$ . These coefficients, together with the constant term  $c$  used to represent the mean of  $w_t$ , are in effect treated as random variables with highly dispersed distributions. This leads to the criterion  $D = M \times S$  again, but with a different value of  $M$  which now depends on the simple input series values  $x_t$ . In the presence of a moderate to large number of simple inputs, the marginal likelihood criterion can counteract bias in the ARIMA model parameters which is caused by estimation of the simple inputs. This is particularly important in relatively short series.

G13BEF can be used with no input series present, to estimate a univariate ARIMA model for the output alone. The marginal likelihood criterion is then distinct from exact likelihood only if a constant term is being estimated in the model, because this is treated as an implicit simple input.

### 3.3 The Estimation Procedure

This is the minimization of the estimation criterion or objective function  $D$  (for deviance). The routine uses an extension of the algorithm of Marquardt (1963). The step size in the minimization is inversely related to a parameter  $\alpha$ , which is increased or decreased by a factor  $\beta$  at successive iterations, depending on the progress of the minimization. Convergence is deemed to have occurred if the fractional reduction of  $D$  in successive iterations is less than a value  $\gamma$ , while  $\alpha < 1$ .

Certain model parameters (in fact all excluding the  $\omega$ s) are subject to stability constraints which are checked throughout to within a specified tolerance multiple  $\delta$  of machine accuracy. Using the least-squares criterion, the minimization may halt prematurely when some parameters ‘stick’ at a constraint boundary. This can happen particularly with short seasonal series (with a small number of whole seasons). It will not happen using the exact likelihood criterion, although convergence to a point on the boundary may sometimes be rather slow, because the criterion function may be very flat in such a region. There is also a smaller risk of a premature halt at a constraint boundary when marginal likelihood is used.

A positive, or zero number of iterations can be specified. In either case, the value  $D$  of the objective function at iteration zero is presented at the initial parameter values, except for estimation of any pre-period terms for the input series, backforecasts for the noise series, and the coefficients of any simple inputs, and the constant term (unless this is held fixed).

At any later iteration, the value of  $D$  is supplied after re-estimation of the backforecasts to their optimal values, corresponding to the model parameters presented at that iteration. This is not true for any pre-period terms for the input series which, although they are updated from the previous iteration, may not be precisely optimal for the parameter values presented, unless convergence of those parameters has occurred. However, in the case of marginal likelihood being specified, the coefficients of the simple inputs and the constant term are also re-estimated together with the backforecasts at each iteration, to values which are optimal for the other parameter values presented.

### 3.4 Further Results

The residual variance is taken as  $\text{erv} = \frac{S}{\text{df}}$ , where  $\text{df} = N - (\text{total number of parameters estimated})$ , is the residual degrees of freedom. The pre-period nuisance parameters for the input series are included in the reduction of  $\text{df}$ , as is the constant if it is estimated.

The covariance matrix of the vector of model parameter estimates is given by

$$\text{erv} \times H^{-1}$$

where  $H$  is the linearised least-squares matrix taken from the final iteration of the algorithm of Marquardt. From this expression are derived the vector of standard deviations, and the correlation matrix of parameter estimates. These are approximations which are only valid asymptotically, and must be treated with great caution when the parameter estimates are close to their constraint boundaries.

The residual series  $a_t$  is available upon completion of the iterations over the range  $t = 1 + d + s \times D, \dots, n$  corresponding to the differenced noise series  $w_t$ .

Because of the algorithm used for backforecasting, these are only true residuals for  $t \geq 1 + q + s \times Q - p - s \times P - d - s \times D$ , provided this is positive. Estimation of pre-period terms for the inputs will also tend to reduce the magnitude of the early residuals, sometimes severely.

The model component series  $z_{1,t}, \dots, z_{m,t}$  and  $n_t$  may optionally be returned in place of the supplied series values, in order to assess the effects of the various inputs on the output.

### 3.5 Forecasting Information

For the purpose of constructing forecasts of the output series at future time points  $t = n + 1, n + 2, \dots$  using G13BHF, it is not necessary to use the whole set of observations  $y_t$  and  $x_{1,t}, x_{2,t}, \dots, x_{m,t}$  for  $t = 1, 2, \dots, m$ . It is sufficient to retain a limited set of quantities constituting the 'state set' as follows: for each series which appears with lagged subscripts in equations (a), (b), (c), (d) and (e) above, include the values at times  $n + 1 - k$  for  $k = 1$  up to the maximum lag associated with that series in the equations. Note that (c) implicitly includes past values of  $n_t$  and intermediate differences of  $n_t$  such as  $\nabla^{d-1} \nabla_s^D$ .

If later observations of the series become available, it is possible to update the state set (without re-estimating the model) using G13BGF. If time series data is supplied with a previously estimated model, it is possible to construct the state set (and forecasts) using G13BJF.

## 4 References

Box G E P and Jenkins G M (1976) *Time Series Analysis: Forecasting and Control* (Revised Edition) Holden-Day

Marquardt D W (1963) An algorithm for least-squares estimation of nonlinear parameters *J. Soc. Indust. Appl. Math.* **11** 431

## 5 Parameters

1: MR(7) – INTEGER array *Input*

*On entry:* the orders vector  $(p, d, q, P, D, Q, s)$  of the ARIMA model for the output noise component.

$p, q, P$  and  $Q$  refer respectively to the number of autoregressive ( $\phi$ ), moving average ( $\theta$ ), seasonal autoregressive ( $\Phi$ ) and seasonal moving average ( $\Theta$ ) parameters.

$d, D$  and  $s$  refer respectively to the order of non-seasonal differencing, the order of seasonal differencing and the seasonal period.

2: NSER – INTEGER *Input*

*On entry:* the total number of input and output series. There may be any number of input series (including none), but always one output series.

*Constraints:*

NSER > 1 if there are no parameters in the model (that is,  $p = q = P = Q = 0$  and KFC = 0),

NSER  $\geq$  1 otherwise.

3: MT(4,NSER) – INTEGER array *Input*

*On entry:* the transfer function model orders  $b, p$  and  $q$  of each of the input series. The order parameters for input series  $i$  are held in column  $i$ . Row 1 holds the value  $b_i$ , row 2 holds the value  $q_i$  and row 3 holds the value  $p_i$ . For a simple input,  $b_i = q_i = p_i = 0$ .

Row 4 holds the value  $r_i$ , where  $r_i = 1$  for a simple input,  $r_i = 2$  for a transfer function input for which no allowance is to be made for pre-observation period effects, and  $r_i = 3$  for a transfer function input for which pre-observation period effects will be treated by estimation of appropriate nuisance parameters.

When  $r_i = 1$ , any non-zero contents of rows 1, 2, and 3 of column  $i$  are ignored.

- 4: PARA(NPARA) – *real* array *Input/Output*  
*On entry:* initial values of the multi-input model parameters. These are in order, firstly the ARIMA model parameters:  $p$  values of  $\phi$  parameters,  $q$  values of  $\theta$  parameters,  $P$  values of  $\Phi$  parameters and  $Q$  values of  $\Theta$  parameters. These are followed by initial values of the transfer function model parameters  $\omega_0, \omega_1, \dots, \omega_{q_i}, \delta_1, \delta_2, \dots, \delta_{p_i}$  for the first of any input series and similarly for each subsequent input series. The final component of PARA is the initial value of the constant  $c$ , whether it is fixed or is to be estimated.  
*On exit:* the latest values of the estimates of these parameters.
- 5: NPARA – INTEGER *Input*  
*On entry:* the exact number of  $\phi, \theta, \Phi, \Theta, \omega, \delta$  and  $c$  parameters.  
*Constraint:*  $NPARA = p + q + P + Q + NSER + \sum(p_i + q_i)$ , the summation being over all the input series.  $c$  must be included, whether fixed or estimated.
- 6: KFC – INTEGER *Input*  
*On entry:* KFC must be set to 0 if the constant  $c$  is to remain fixed at its initial value, and 1 if it is to be estimated.  
*Constraint:*  $KFC = 0$  or  $1$ .
- 7: NXXY – INTEGER *Input*  
*On entry:* the (common) length of the original, undifferenced input and output time series.
- 8: XXY(IXXY,NSER) – *real* array *Input/Output*  
*On entry:* the columns of XXY must contain the NXXY original, undifferenced values of each of the input series and the output series in that order.  
*On exit:* if  $KZEF = 0$ , XXY remains unchanged on exit.  
 If  $KZEF \neq 0$ , the columns of XXY hold the corresponding values of the input component series  $z_t$  in place of  $x_t$  and the output noise component  $n_t$  in place of  $y_t$ , in that order.
- 9: IXXY – INTEGER *Input*  
*On entry:* the first dimension of the array XXY as declared in the (sub)program from which G13BEF is called.  
*Constraint:*  $IXXY \geq NXXY$ .
- 10: KEF – INTEGER *Input*  
*On entry:* indicates the likelihood option.  $KEF = 1$  gives least-squares;  $KEF = 2$  gives exact likelihood;  $KEF = 3$  gives marginal likelihood.  
*Constraint:*  $KEF = 1, 2$  or  $3$ .
- 11: NIT – INTEGER *Input*  
*On entry:* the maximum required number of iterations. If  $NIT = 0$ , no change is made to any of the model parameters in array PARA except that the constant  $c$  (if  $KFC = 1$ ) and any  $\omega$  relating to simple input series are estimated. (Apart from these, estimates are always derived for the nuisance parameters relating to any backforecasts and any pre-observation period effects for transfer function inputs.)  
*Constraint:*  $NIT \geq 0$ .
- 12: KZSP – INTEGER *Input*  
*On entry:* KZSP must be set to 1 if the routine is to use the input values of ZSP in the minimization procedure, and to any other value if the default values of ZSP are to be used.

- 13: ZSP(4) – *real* array *Input/Output*
- On entry:* if KZSP = 1, then ZSP must contain the four values used to control the strategy of the search procedure. These are as follows.
- ZSP(1) contains  $\alpha$ , the value used to constrain the magnitude of the search procedure steps.
- ZSP(2) contains  $\beta$ , the multiplier which regulates the value of  $\alpha$ .
- ZSP(3) contains  $\delta$ , the value of the stationarity and invertibility test tolerance factor.
- ZSP(4) contains  $\gamma$ , the value of the convergence criterion.
- If KZSP  $\neq$  1 before entry, default values of ZSP are supplied by the routine. These are 0.01, 10.0, 1000.0 and  $\max(100 \times \textit{machine precision}, 0.0000001)$ , respectively.
- On exit:* contains the values, default or otherwise, used by the routine.
- Constraints:*
- if KZSP = 1, then  
 ZSP(1) > 0.0,  
 ZSP(2) > 1.0,  
 ZSP(3)  $\geq$  1.0,  
 $0 \leq$  ZSP(4) < 1.0.
- 14: ITC – INTEGER *Output*
- On exit:* the number of iterations carried out.
- A value of ITC = –1 on exit indicates that the only estimates obtained up to this point have been for the nuisance parameters relating to backforecasts, unless the marginal likelihood option is used, in which case estimates have also been obtained for simple input coefficients  $\omega$  and for the constant  $c$  (if KFC = 1). This value of ITC usually indicates a failure in a consequent step of estimating transfer function input pre-observation period nuisance parameters.
- A value of ITC = 0 on exit indicates that estimates have been obtained up to this point for the constant  $c$  (if KFC = 1), for simple input coefficients  $\omega$  and for the nuisance parameters relating to the backforecasts and to transfer function input pre-observation period effects.
- 15: SD(NPARA) – *real* array *Output*
- On exit:* the NPARA values of the standard deviations corresponding to each of the parameters in PARA. When the constant is fixed its standard deviation is returned as zero. When the values of PARA are valid, the values of SD are usually also valid. However, if an exit value of IFAIL = 3, 8 or 10 is accompanied by a failure to invert the second derivative matrix (which would normally give IFAIL = 9), then the contents of SD will be indeterminate.
- 16: CM(ICM,NPARA) – *real* array *Output*
- On exit:* the first NPARA rows and columns of CM contain the correlation coefficients relating to each pair of parameters in PARA. All coefficients relating to the constant will be zero if the constant is fixed. The contents of CM will be indeterminate under the same conditions as SD.
- 17: ICM – INTEGER *Input*
- On entry:* the first dimension of the array CM as declared in the (sub)program from which G13BEF is called.
- Constraint:* ICM  $\geq$  NPARA.
- 18: S – *real* *Output*
- On exit:* the residual sum of squares,  $S$ , at the latest set of valid parameter estimates.
- 19: D – *real* *Output*
- On exit:* the objective function,  $D$ , at the latest set of valid parameter estimates.

- 20: NDF – INTEGER Output  
*On exit:* the number of degrees of freedom associated with  $S$ .
- 21: KZEF – INTEGER Input  
*On entry:* KZEF must not be set to 0, if the values of the input component series  $z_t$  and the values of the output noise component  $n_t$  are to overwrite the contents of XXY on exit, and must be set to 0 if XXY is to remain unchanged.
- 22: RES(NXXY) – *real* array Output  
*On exit:* the values of the residuals relating to the differenced values of the output series. The remainder of the first NXXY terms in the array will be zero.
- 23: STTF(ISTTF) – *real* array Output  
*On exit:* the NSTTF values of the state set array.
- 24: ISTTF – INTEGER Input  
*On entry:* the dimension of the array STTF as declared in the (sub)program from which G13BEF is called.  
*Constraint:*  $ISTTF \geq (P \times s) + d + (D \times s) + q + \max(p, Q \times s) + ncg$ , where  $ncg = \sum (b_i + q_i + p_i)$  over all input series for which  $r_i > 1$ .
- 25: NSTTF – INTEGER Output  
*On exit:* the number of values in the state set array STTF.
- 26: WA(IWA) – *real* array Workspace  
 27: IWA – INTEGER Input  
*On entry:* the dimension of the array WA as declared in the (sub)program from which G13BEF is called.  
 It is not practical to outline a method for deriving the exact minimum permissible value of IWA, but the following gives a reasonably good conservative approximation. (It should be noted that if IWA is too small (but not grossly so) then the exact minimum is returned in MWA( $i$ ) and is also printed if KPRIV  $\neq$  0.)  
 Let  $q' = q + (Q \times s)$  and  $d' = d + (D \times s)$  where the orders of the output noise model are  $p, d, q, P, D, Q, s$ .  
 Let there be  $l$  input series, where  $l = NSER - 1$ .  
 Let  $mx_i = \max(b_i + q_i, p_i)$ , if  $r_i = 3$  for  $i = 1, 2, \dots, l$   
 $mx_i = 0$ , if  $r_i \neq 3$  for  $i = 1, 2, \dots, l$   
 where the transfer function model orders for input  $i$  are given by  $b_i, q_i, p_i, r_i$ .  
 Let  $qx = \max(q', mx_1, mx_2, \dots, mx_l)$ .  
 Let  $ncd = NPARA + KFC + qx + \sum_{i=1}^l mx_i$  and  $nce = NXXY + d' + 6 \times qx$ .  
 Finally, let  $ncf = NSER$ , and then increment  $ncf$  by 1 every time any of the following conditions is satisfied. (The last six conditions should be applied separately to each input series, so that, for example, if we have two input series and if  $p_1 > 0$  and  $p_2 > 0$  then  $ncf$  is incremented by 2.)

The conditions are:

$$\begin{aligned} p &> 0 \\ q &> 0 \\ P &> 0 \\ Q &> 0 \\ qx &> 0 \\ \text{KFC} &> 0 \end{aligned}$$

$$\left. \begin{aligned} p &> 0 \\ q &> 0 \\ P &> 0 \\ Q &> 0 \end{aligned} \right\} \text{and } q > 0 \text{ and } \text{KEF} > 1.$$

$$\left. \begin{aligned} p &> 0 \\ q &> 0 \\ P &> 0 \\ Q &> 0 \end{aligned} \right\} \text{and } \text{KFC} > 0 \text{ and } \text{KEF} = 3.$$

$$\left. \begin{aligned} mx_i &> 0 \\ p_i &> 0 \\ p &> 0 \\ q &> 0 \\ P &> 0 \\ Q &> 0 \end{aligned} \right\} \text{and } r_i = 1 \text{ and } \text{KEF} > 3 \text{ separately for } i = 1, 2, \dots, l.$$

Then  $\text{IWA} \geq 2 \times (\text{ncd})^2 + (\text{ncc}) \times (\text{ncf} + 4)$ .

- 28: MWA(IMWA) – INTEGER array  
29: IMWA – INTEGER

*Workspace*  
*Input*

*On entry:* the dimension of the array MWA as declared in the (sub)program from which G13BEF is called.

*Constraint:*  $\text{IMWA} \geq (16 \times \text{NSER}) + (7 \times \text{ncd}) + (3 \times \text{NPARA}) + (3 \times \text{KFC}) + 27$ , where the derivation of *ncd* is shown under IWA above.

If IMWA is too small then the exact minimum needed is returned in IMWA and if KPRIV  $\neq 0$  it is also printed.

- 30: KPRIV – INTEGER

*Input*

*On entry:* KPRIV must not be set to 0, if it is required to monitor the course of the optimization or to print out the requisite minimum values of IWA or IMWA in the event of an error of the type IFAIL = 6 or 7. The course of the optimization is monitored by printing out at each iteration the iteration count (ITC), the residual sum of squares (S), the objective function (D) and a description and value for each of the parameters in the PARA array. The descriptions are PHI for  $\phi$ , THETA for  $\theta$ , SPHI for  $\Phi$ , STHETA for  $\Theta$ , OMEGA/SI for  $\omega$  in a simple input, OMEGA for  $\omega$  in a transfer function input, DELTA for  $\delta$  and CONSTANT for  $c$ . In addition SERIES 1, SERIES 2, etc. indicate the input series relevant to the OMEGA and DELTA parameters.

KPRIV must be set to 0 if the print-out of the above information is not required.



31: IFAIL – INTEGER

*Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL  $\neq$  0 on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, KFC < 0,  
 or KFC > 1,  
 or IXXY < NXXY,  
 or ICM < NPARA,  
 or KEF < 1,  
 or KEF > 3,  
 or NIT < 0,  
 or NSER < 1,  
 or NSER = 1 and there are no parameters in the model ( $p = q = P = Q = 0$  and KFC = 0).

IFAIL = 2

On entry, there is inconsistency between NPARA and KFC on the one hand and the orders in arrays MR and MT on the other,  
 or one of the  $r_i$ , stored in MT(4,  $i$ ), is not equal to 1, 2 or 3.

IFAIL = 3

On entry or during execution, one or more sets of  $\delta$  parameters do not satisfy the stationarity or invertibility test conditions.

IFAIL = 4

On entry, when KZSP = 1, ZSP(1)  $\leq$  0.0,  
 or ZSP(2)  $\leq$  1.0,  
 or ZSP(3) < 1.0,  
 or ZSP(4) < 0.0,  
 or ZSP(4)  $\geq$  1.0.

IFAIL = 5

On entry, IWA is too small by a considerable margin. No information is supplied about the requisite minimum size.

IFAIL = 6

On entry, IWA is too small, but the requisite minimum size is returned in MWA(1), which is printed if KPRIV  $\neq$  0.

IFAIL = 7

On entry, IMWA is too small, but the requisite minimum size is returned in MWA(1), which is printed if KPRIV  $\neq$  0.

IFAIL = 8

This indicates a failure in F04ASF which is used to solve the equations giving the latest estimates of the parameters.

IFAIL = 9

This indicates a failure in the inversion of the second derivative matrix. This is needed in the calculation of the correlation matrix and the standard deviations of the parameter estimates.

IFAIL = 10

On entry or during execution, one or more sets of the ARIMA ( $\phi$ ,  $\theta$ ,  $\Phi$  or  $\Theta$ ) parameters do not satisfy the stationarity or invertibility test conditions.

IFAIL = 11

On entry, ISTTF is too small. The state set information will not be produced and if KZEF  $\neq$  0 array XXY will remain unchanged. All other parameters will be produced correctly.

IFAIL = 12

The routine has failed to converge after NIT iterations. If steady decreases in the objective function,  $D$ , were monitored up to the point where this exit occurred, then the exit probably occurred because NIT was set too small, so the calculations should be restarted from the final point held in PARA.

IFAIL = 13

On entry, ISTTF is too small (see IFAIL = 11) and NIT iterations were carried out without the convergence conditions being satisfied (see IFAIL = 12).

## 7 Accuracy

The computation used is believed to be stable.

## 8 Further Comments

The time taken by the routine is approximately proportional to  $NXXY \times ITC \times NPARA^2$ .

## 9 Example

The data in the example relates to 40 observations of an output time series and of a single input time series. The noise series has one autoregressive ( $\phi$ ) and one seasonal moving average ( $\Theta$ ) parameter (both of which are initially set to zero) for which the seasonal period is 4. The input series is defined by orders  $b_1 = 1$ ,  $q_1 = 0$ ,  $p_1 = 1$ ,  $r_1 = 3$ , so that it has one  $\omega$  (initially set to 2.0) and one  $\delta$  (initially set to 0.5), and allows for pre-observation period effects. The constant (initially set to zero) is to be estimated. Default values of ZSP are used. Up to 20 iterations are allowed, and the progress of these is not monitored. Marginal likelihood is the chosen option.

After the full 11 iterations, the following are computed and printed out: the final values of the PARA parameters and their standard errors, the correlation matrix, the residuals for the 36 differenced values, the values of  $z_t$  and  $n_t$ , the values of the state set and the number of degrees of freedom.

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G13BEF Example Program Text
*      Mark 15 Revised.  NAG Copyright 1991.
*      .. Parameters ..
INTEGER      NSERMX, NPMAX, NXXYMX, ISTTF, IXXY, ICM, IWA,
+           IMWA
PARAMETER    (NSERMX=2, NPMAX=10, NXXYMX=50, ISTTF=20,
+           IXXY=NXXYMX, ICM=NPMAX, IWA=1500, IMWA=200)
INTEGER      NIN, NOUT
PARAMETER    (NIN=5, NOUT=6)
*      .. Local Scalars ..
real       D, S
INTEGER      I, IFAIL, ITC, J, KEF, KFC, KPRIV, KZEF, KZSP,
+           NDF, NDV, NIT, NPARA, NSER, NSTTF, NXXY
*      .. Local Arrays ..
real       CM(ICM, NPMAX), PARA(NPMAX), RES(NXXYMX),
+           SD(NPMAX), STTF(ISTTF), WA(IWA),
+           XXY(IXXY, NSERMX), ZSP(4)
INTEGER      MR(7), MT(4, NSERMX), MWA(IMWA)
*      .. External Subroutines ..
EXTERNAL     G13BEF, X04ABF
*      .. Executable Statements ..
WRITE (NOUT,*) 'G13BEF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) KZEF, KFC, NXXY, NSER, KEF, NIT, KZSP
IF (NXXY.GT.0 .AND. NXXY.LE.NXXYMX .AND. NSER.GT.0 .AND. NSER.LE.
+   NSERMX) THEN
  IF (KZSP.NE.0) READ (NIN,*) ZSP
  CALL X04ABF(1, NOUT)
  READ (NIN,*) (MR(I), I=1, 7)
  DO 20 I = 1, 4
    READ (NIN,*) (MT(I, J), J=1, NSER)
20  CONTINUE
  NPARA = 0
  DO 40 I = 1, NSER
    NPARA = NPARA + MT(2, I) + MT(3, I)
40  CONTINUE
  NPARA = NPARA + MR(1) + MR(3) + MR(4) + MR(6) + NSER
  IF (NPARA.LE.NPMAX) THEN
    READ (NIN,*) (PARA(I), I=1, NPARA)
    DO 60 I = 1, NXXY
      READ (NIN,*) (XXY(I, J), J=1, NSER)
60  CONTINUE
*      * Set KPRIV to 1 to obtain monitoring information *
      KPRIV = 0
      IFAIL = 1
*
      CALL G13BEF(MR, NSER, MT, PARA, NPARA, KFC, NXXY, XXY, IXXY, KEF, NIT,
+             KZSP, ZSP, ITC, SD, CM, ICM, S, D, NDF, KZEF, RES, STTF,
+             ISTTF, NSTTF, WA, IWA, MWA, IMWA, KPRIV, IFAIL)
*
      IF (IFAIL.NE.0) THEN
        WRITE (NOUT,*)
        WRITE (NOUT,99999) 'G13BEF fails. IFAIL =', IFAIL
      END IF
      IF (IFAIL.EQ.0 .OR. IFAIL.EQ.8 .OR. IFAIL.EQ.9 .OR.
+       IFAIL.EQ.11) THEN
        WRITE (NOUT,*)
        WRITE (NOUT,99999)
+       'The number of iterations carried out is', ITC
        WRITE (NOUT,*)
        WRITE (NOUT,*)
+       'Final values of the parameters and their standard deviations'
        WRITE (NOUT,*)
        WRITE (NOUT,*)

```

```

+       ' I          PARA(I)          SD'
      WRITE (NOUT,*)
      DO 80 I = 1, NPARA
        WRITE (NOUT,99998) I, PARA(I), SD(I)
80     CONTINUE
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'The correlation matrix is'
      WRITE (NOUT,*)
      WRITE (NOUT,99997) ((CM(I,J),J=1,NPARA),I=1,NPARA)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'The residuals and the z and n values are'
      WRITE (NOUT,*)
+       ' I          RES(I)          z(t)          n(t)'
      WRITE (NOUT,*)
      NDV = NXXY - MR(2) - MR(5)*MR(7)
      DO 100 I = 1, NXXY
        IF (I.LE.NDV) THEN
          WRITE (NOUT,99996) I, RES(I), (XXY(I,J),J=1,NSER)
        ELSE
          WRITE (NOUT,99995) I, (XXY(I,J),J=1,NSER)
100     CONTINUE
        IF (MR(2).NE.0 .OR. MR(5).NE.0) THEN
          WRITE (NOUT,*)
          WRITE (NOUT,*)
+       '** Note that the residuals relate to differenced values **'
          END IF
          WRITE (NOUT,*)
          WRITE (NOUT,99994) 'The state set consists of', NSTTF,
+       ' values'
          WRITE (NOUT,*)
          WRITE (NOUT,99993) (STTF(I),I=1,NSTTF)
          WRITE (NOUT,*)
          WRITE (NOUT,99999) 'The number of degrees of freedom is',
+       NDF
        END IF
      END IF
    END IF
  END IF
  STOP
*
99999 FORMAT (1X,A,I4)
99998 FORMAT (1X,I4,2F20.6)
99997 FORMAT (1X,5F10.4)
99996 FORMAT (1X,I4,3F15.3)
99995 FORMAT (1X,I4,F30.3,F15.3)
99994 FORMAT (1X,A,I4,A)
99993 FORMAT (1X,6F10.4)
END

```

## 9.2 Program Data

G13BEF Example Program Data

1	1	40	2	3	20	0
1	0	0	0	0	1	4
1	0					
0	0					
1	0					
3	0					
0.0	0.0	2.0	0.5	0.0		
8.075	105.0					
7.819	119.0					
7.366	119.0					
8.113	109.0					
7.380	117.0					
7.134	135.0					
7.222	126.0					
7.768	112.0					
7.386	116.0					
6.965	122.0					

6.478	115.0
8.105	115.0
8.060	122.0
7.684	138.0
7.580	135.0
7.093	125.0
6.129	115.0
6.026	108.0
6.679	100.0
7.414	96.0
7.112	107.0
7.762	115.0
7.645	123.0
8.639	122.0
7.667	128.0
8.080	136.0
6.678	140.0
6.739	122.0
5.569	102.0
5.049	103.0
5.642	89.0
6.808	77.0
6.636	89.0
8.241	94.0
7.968	104.0
8.044	108.0
7.791	119.0
7.024	126.0
6.102	119.0
6.053	103.0

### 9.3 Program Results

G13BEF Example Program Results

The number of iterations carried out is 11

Final values of the parameters and their standard deviations

I	PARA(I)	SD
1	0.380924	0.166379
2	-0.257786	0.178178
3	8.956084	0.948061
4	0.659641	0.060239
5	-75.435521	33.505341

The correlation matrix is

1.0000	-0.1839	-0.1775	-0.0340	0.1394
-0.1839	1.0000	0.0518	0.2547	-0.2860
-0.1775	0.0518	1.0000	-0.3070	-0.2926
-0.0340	0.2547	-0.3070	1.0000	-0.8185
0.1394	-0.2860	-0.2926	-0.8185	1.0000

The residuals and the z and n values are

I	RES(I)	z(t)	n(t)
1	0.397	180.567	-75.567
2	3.086	191.430	-72.430
3	-2.818	196.302	-77.302
4	-9.941	195.460	-86.460
5	-5.061	201.594	-84.594
6	14.053	199.076	-64.076
7	2.624	195.211	-69.211
8	-5.823	193.450	-81.450
9	-2.147	197.179	-81.179
10	-0.216	196.217	-74.217
11	-2.517	191.812	-76.812

12	7.916	184.544	-69.544
13	1.423	194.322	-72.322
14	11.936	200.369	-62.369
15	5.117	200.990	-65.990
16	-5.672	200.468	-75.468
17	-5.681	195.763	-80.763
18	-1.637	184.025	-76.025
19	-1.019	175.360	-75.360
20	-2.623	175.492	-79.492
21	3.283	182.162	-75.162
22	6.896	183.857	-68.857
23	5.395	190.797	-67.797
24	0.875	194.327	-72.327
25	-4.153	205.558	-77.558
26	6.206	204.261	-68.261
27	4.208	207.104	-67.104
28	-2.387	196.423	-74.423
29	-11.803	189.924	-87.924
30	6.435	175.158	-72.158
31	1.342	160.761	-71.761
32	-4.924	156.575	-79.575
33	4.799	164.256	-75.256
34	-0.074	167.783	-73.783
35	-6.023	184.483	-80.483
36	-6.427	193.055	-85.055
37	-2.527	199.390	-80.390
38	2.039	201.302	-75.302
39	0.243	195.695	-76.695
40	-3.166	183.738	-80.738

The state set consists of 6 values

6.0530 183.7384 -5.7855 -0.1645 0.1800 -3.0977

The number of degrees of freedom is 34

---